Упражнение 6

Запознаване със средата. Основни команди, конструкции, типове данни, факти, правила

Основни команди в CLIPS:

(exit) - Затваря CLIPS.

(clear) - Премахва всички правила и факти от паметта на CLIPS. Еквивалентно на затваряне и рестариране на CLIPS.

(load <файл.clp>) – Зарежда конструкциите в посочения файл

(reset) - Премахва информацията за всички факти от паметта (но не и правилата) и изчиства конфликтното множество.

(run) - Стартира изпълнението на програмата.

В CLIPS се предоставят осем примитивни типа данни за представяне на информацията. Тези типове са: float, integer, symbol, string, externaladdress, factaddress, instancename, instanceaddress. Числовите данни да се представят чрез float и integer. Символните данни чрез symbol и string. number се състои само от цифри (0-9), десетична точка (.), знак (+ или -) и незадължителен символ (е) за експоненциална нотация със съответен знак. Числата се представят или чрез float или чрез integer. Всяко число може да има знак следван само от цифри представено като integer (вътрешно се представя в CLIPS като C long integer). Всички други числа се представят като float (вътрешно се представят като C double float). Броят на значещите цифри зависи от машинната реализация. Възможно е да се получат грешки от закръгляне, което също зависи от реализацията. Както във всеки друг програмен език така и в CLIPS трябва да се внимава при извършване на сравнение между стойности от различни типове. Примери за integer стойности са:

237
15
+12
-32

Примери за float стойности са:

237e3

15.09

+12.0

-32.3e-7

integer използва следния формат:

<integer> ::= [+ | -] <digit>+

<digit> ::= 0 | 1 | 2 | 3 | 4 | 5 | 6 | 7 | 8 | 9

float числата имат формат:

<float> ::= <integer> <exponent> |

            <integer> . [exponent]

            . <unsigned integer> [exponent]

            <integer> . <unsigned integer> [exponent]

<unsigned-integer> ::= <digit>+

<exponent> ::= e | E <integer>

Последователност от символи, които не съвпадат с формата на числовите типове се разглеждат като symbol.

symbol в CLIPS е всяка поредица от символи които започват с някакъв ASCII символ, който може да бъде отпечатан следван от нула или повече също такива символи. symbol завършва при срещане на разделител. Разделители са: ASCII символ, който не може да бъде разпечатан (включително интервал, табулация, връщане на каретката и преминаване на следващ ред), двойни кавички, отварящи и затварящи скоби, знак &, вертикална черта (|), по-малко от (<) и тилда (~). С точка и запетайка (;) се отбелязва коментар и също може да се разглежда като разделител. Разделителите не могат да се включват в типа symbol с изключение на  символа (<), който може да е първия символ в symbol типа. Освен това symbol не може да започва с ? или $? (въпреки че може да съдържа тези символи). Тези символи се използват за означаване на променливи. CLIPS е чувствителен към регистъра (малките и големи букви се различават). Важно е да се отбележи, че числата са специален случай на symbol (т.е. те удовлетворяват дефиницията на symbol, но се разглеждат като различен тип данни). Някой примери за symbol са:

Foo
Hello

B76-HI
bad_value

127A
456-93-039
@+=-%

2each
string е множество от символи започващи с двойни кавички следвани от нула или повече символи които могат да бъдат разпечатани. string завършва с двойни кавички. В string могат да се използват двойни кавички предхождани от обратна наклоненна черта (\). Обратна наклонена черта може да се използва чрез две обратни наклонени черти. Примери за string:

"foo"

"a and b"
"1 number"
"a\"quote"

string стойността “abcd” не е същото като symbol стойността abcd. Те съдържат същите символи, но са от различни типове. Тя обаче е същата за instance name [abcd].

external address е адрес на външна даннова структура върната от функция (написана на език като C или Ada) интегрирана в CLIPS. Този тип данни може да се създаде само чрез извикване на функция (т.е. не е възможно да се специфицира external address чрез задаване на стойност). В базовата версия на CLIPS (в която няма потребителски дефинирани външни функциии), не е възможно да се създаде такъв тип данни. В CLIPS печатната форма на external address е:

<Pointer-XXXXXX>

където XXXXXX е external-address
Факт е списък от атомарни стойности, които са представени или позиционно (наредени факти) или по име (ненаредени факти или темплейти). Фактите се представят чрез индекс или адрес. Печатния формат на адрес на факт е:

<Fact-XXX>

където XXX е индекс на факт.

instance е обект който е инстанция на определен клас. Обекти в CLIPS са float, integer, symbol, string, multifield value, externaladdress, factaddress или инстанции на потребителски дефинирани класове. Клас се създава чрез defclass кунструкцията. Инстанция на потребителски дефиниран клас се създава чрез функцията makeinstance и такава инстанция се определя уникално чрез адрес. В обхвата на модула инстанцията може да се определя уникално и чрез име. instancename се формира чрез заграждане в квадратни скоби на symbol. Чистите symbol не трябва да се заграждат в квадратни скоби. Примери:

[pump-1]
[foo]

[+++]

[123-890]

Скобите на са част от името на инстанцията, те просто служат за означаване че затвореният в скоби symbol е instancename. instanceaddress може да се получи единствено чрез свързване на стойност на функция наречена instanceaddress или чрез свързване на променлива към инстанция съответстваща на обект образец в лявата част на правило (т.е. възможно е да се зададе instanceaddress чрез ръчно записване на стойност). В CLIPS печатния формат на instanceaddress е:

<Instance-XXX>

където XXX е името на инстанцията.

В CLIPS място на което има стойност (някой от примитивните типове данни) се нарича поле (field). Примитивните типове данни се наричат singlefield values. Константа (constant) е непроменяща се single field стойност записана като последователност от символи (което означва, че externaladdress, factaddress и instanceaddress не могат да се изразят като константи, тъй като те могат да се получат само чрез функционално извикване или свързване на променливи). multifield value е поредица от нула или повече single field values. Когато се изведат на екран от CLIPS, multifield values се заграждат в скоби. Общо single и multifield values се наричат values. Някой примери за multifield values са:

(a)
(1 bar foo)
()
(x 3.0 "red" 567)

multifield стойността (а) не е същото като single field стойност а. multifield value се създават или чрез извикване на функции които връщат multifield стойности, чрез използване на wildcard аргумент в deffunction, object messagehandler или method или чрез свързване на променливи по време на проверката за удовлетворимост на правилата. В CLIPS, variable е именувано място използвано за съхраняване на стойности. Променливите се използват от много CLIPS конструкции (като defrule, deffunction, defmethod и defmessagehandler).

Най-общо CLIPS оперира чрез поддържане на списък от факти и множество от правила които действат върху тях. Фактът е данна като (color green) или (parent_of John Susan). Фактите се създават чрез добавянето им в базата от данни за факти с използване на командата (assert). Пример с отговор от CLIPS:

CLIPS>(assert (colour green))

<Fact-0>

<Fact-0> частта е отговорът от които CLIPS показва че се създава нов факт (факт номер 0) и той се помества в базата от данни за факти. Командата (facts) извежда всички текущи факти. Резултатът от изпълнението и в случая ще бъде:

CLIPS>(facts)

f-0 (colour green)

For a total of 1 fact.

Фактите могат да се премахват от базата от данни чрез командата (retract). Например след добавянето на два факта по следния начин:

CLIPS>(assert (colour green))

<Fact-0>

CLIPS>(assert (colour red))

<Fact-1>

Премахването на първия факт и извеждането на списъка от факти ще даде като резултат:

CLIPS>(retract 0)

CLIPS>(facts)

f-1 (colour red)

For a total of 1 fact.

Има две важни неща които трябва да се отбележат: първо за да се премахне даден факт трябва да се укаже номер (индекс на факт), не самия факт и второ индексите на факти не могат да се използват повторно. След като веднъж факт 0 е премахнат, следващият добавен такъв ще има индекс 2, а не 0.

Самите факти нямат широко използване. Програмата трябва да извършва някакви действия, чрез създаването на правила. Най-общо едно правило се изразява във вида 'IF нещо е истина THEN изпълни някакво действие'. Тези правила са известни като продукции. Поради това експертните системи базирани на правила често се наричат продукционни системи. (CLIPS всъщност означава C Language Integrated Production System). В CLIPS едно правило обикновено има следния вид:

(defrule duck

  (animal-is duck)

  =>

  (assert (sound-is quack)))

Правилото се състои от три части. Първата част (defrule duck) просто задава уникално име на правило. Втората част (animal-is duck) е образец (IF частта) на правилото и последната част (assert (sound-is quack)) е действието (THEN частта). При конвенционалните езици това правило означава: 'ако има факт (animal-is duck) в базата от данни за факти, то добави друг факт (sound-is quack) в тази база'. Ако се изпълни командата (rules) ще се изведе списък от правила (в случая точно едно) представени в системата. Сега ако се изпълни командата (assert (animal-is duck)) и се провери списъка от факти ще се изведе един факт. За да сработи правилото трябвада се изпълни командата (run). Въпреки че изглежда нищо не се е случило ако отново се провери списъка с факти ще се види че има нов факт (sound-is quack), който е свързан с изпълнението на правилото. Това е силата на програмирането базирано на правила - способността да се въздейства от данните, така резултата от едно правило може да се използва като образец за друго. След добавяне на правилото:

(defrule is-it-a-duck

  (animal-has webbed-feet)

  (animal-has feathers)

  =>

  (assert (animal-is duck)))

и изпълнение на командата (reset) за премахване на фактите правилата няма да се променят. Това правило има два образеца. За да се изпълни действието е необходимо и двата да бъдат удовлетворени. Това има следния смисъл: 'IF животното има ципи на краката AND животното има пера THEN животното е патица'. Сега ако се добавят фактите (animal-has webbed-feet) и (animal-has feathers) ще има два представени факта. След изпълнение на командата (run) ще има четири факта. Първо сработва правилото is-it-a-duck, добавящо факта (animal-is duck). След това този факт удовлетворява правилото duck, което добавя факта (sound-is quack). Използвайки тази възможност за верижно изпълнение на правила могат да се построят много мощни системи.

Добавянето на факти по-скоро е неудобен начин за представяне на резултати. Първото правило може да се запише отново, този път с няколко действия (няколко операции в дясната част на правилото):

(defrule duck

  (animal-is duck)

  =>

  (assert (sound-is quack))

  (printout t "it’s a duck" crlf))

След изпълнението на правилата ще се изведе съобщение и ще се добави нов факт.

Не е ефективно всеки път да се написват всички правила когато се стартира (run). Те могат да се заредят от файл използвайки команда 'load' от файловото меню при което се задава файл с разширение .clp и чрез опростен редактор могат правилата да се създават и променят. Фактите могат да се записват във файл по същия начин както това се прави от командния ред.

За следващите задачи е необходимо да се използват едни и същи правила няколко пъти. Затова вместо да се написват всеки път, може да се използва структурата deffacts. По този начин се определят факти които се активират всеки път при изпълнение на командата (reset). Например кода

(deffacts startup (animal dog) (animal duck) (animal haddock))

ще добави три факта в базата от данни всеки път когато се изпълни reset. Добавените по този начин факти са същите както всички други - те могат да се премахнат или да се използват в образци на правила - но дори и да се деактивират те отново ще се появят след (reset). По-долу е показан списъка от факти които ше се използват - те се въвеждат чрез CLIPS редактора в deffacts структура и след изпълнение на reset може да се провери списъка от факти за да се види дали съществуват.

(animal dog) (animal cat) (animal duck) (animal turtle) (warm-blooded dog) (warm-blooded cat) (warm-blooded duck) (lays-eggs duck) (lays-eggs turtle) (child-of dog puppy) (child-of cat kitten) (child-of turtle hatchling)

Използваните до сега образци в правилата за удовлетворяване от факти са сравнително прости. Всеки образец се удовлетворява от един определен факт. Чрез използване на въпросителен знак е възможно да се използва сравнение с множество факти изпълнявайки действията в правилата многократно. Например правилото:

(defrule animal

  (animal ?)

  =>

  (printout t "animal found" crlf))

при изпълнение дава следния резултат:

CLIPS>(run)

Animal found

Animal found

Animal found

Animal found

CLIPS>

което показва че е сработило четири пъти, по веднъж за всеки факт съвпадащ с образеца (animal ?). В този образец, символа ? е wildcard. Той съответства с всеки символ. Могат да се използват произволен брой wildcard в даден образец, но първия символ не трябва да е такъв. Така (child-of ? ?) е правилно използван и ще съответства на четири факта, но (? ? hatchling) е неправилно използване.

Променливи в образците.

Простите wildcard са полезни, но не винаги дават достатъчно информация. Ако обаче вместо ? се използва самостоятелно ?var, стойността на ?var може да се използва всеки път когато правилото сработи за получаване на допълнителна информация.

Пример:

(defrule list-animals

(animal ?name)

=>

(printout t ?name " found" crlf))

Това ще даде следният резултат:

CLIPS>(run)

turtle found

duck found

cat found

dog found

CLIPS>

Правилото се удовлетворява от четири факта и всеки път променливата ?name приема стойността на съответното място във факта удовлетворяващ образеца, така че в частта на правилото отнасяща се до действието тази стойност може да се отпечата. Предимството на този подход е очевидно когато се използват два или повече образеца както в следващия пример:

(defrule mammal

  (animal ?name)

  (warm-blooded ?name)

  (not (lays-eggs ?name))

  =>

  (assert (mammal ?name))

  (printout t ?name " is a mammal" crlf))

Тук се използва функцията not за неизпълнение. Резултатът от правилото е:

CLIPS>(run)

cat is a mammal

dog is a mammal

CLIPS>

Освен това може да се запише правилото:

(defrule mammal2

  (mammal ?name)

  (child-of ?name ?young)

  =>

  (assert (mammal ?young))

  (printout t ?young " is a mammal" crlf))

След изпълнението на което е интересно да се проследи списъка от факти.

CLIPS>(run)

kitten is a mammal

puppy is a mammal

CLIPS>

Факти и правила

За да се премахне факт е необходимо да се знае неговият индекс. Факти могат да се премахват и чрез правилата чрез свързването на факти с променливи. Например:

(defrule remove-mammals

  ?fact <- (mammal ?)

  =>

  (printout t "retracting " ?fact crlf)

  (retract ?fact))

В лявата част на правилото, индекса на всеки факт съответстващ на образеца (mammal ?) се приема от променливата ?fact. Това се извършва чрез стрелка сочеща наляво (<-), След изпълнението на правилото, резултатът е (номерата на фактите могат да са други):

CLIPS>(run)

retracting <Fact-13>

retracting <Fact-14>

retracting <Fact-15>

retracting <Fact-16>

CLIPS>

Всички факти mammal са премахнати.

Логически и математически оператори.

Вече бяха разгледани някои правила при които два или повече образци автоматично се свързват с логическо И, което означава че и двата трябва да са истина за да сработи правилото. Също така бе разгледана not функцията, което означава че остава да се разгледа единствено ИЛИ функцията. CLIPS притежава такава функция както в примера по-долу е показано нейното използване.

(defrule take-umbrella

  (or (weather raining)

      (weather snowing))

  =>

  (assert (umbrella required)))

Което означава “if вали дъжд or вали сняг, then вземи чадър”. Важно е да се забележи начина по който се използва or - преди двата аргумента, а не между тях, което е известно като префиксна нотация и всички CLIPS оператори се използват по този начин. Например за да се изрязи сума от две числа в повечето компютърни езици се записва 5 + 7 (известно като инфиксна нотация). В CLIPS изразът се записва (+ 5 7). Следните примери показват извършване на събиране, изваждане, умножение и деление:

CLIPS>(+ 5 7)

12

CLIPS>(- 5 7)

-2

CLIPS>(* 5 7)

35

CLIPS>(/ 5 7)

0.7142857142857143

CLIPS>

Получаване на информация от потребителя.

CLIPS получава информация от потребителя чрез функцията (read). Където и да се срещне (read), програмата изчаква потребителя да напише нещо и след това използва въведената информация. Наприме може да се използва (assert (user-input (read))) в CLIPS. След натискане на return, трябва да се напише нещо (каквото и да е) преди да завърши командата. Когато се погледнат фактите, ще се забележи, че е добавен нов факт с втори елемент въведената информация от клавиатурата. След изпълнение на правилото:

(defrule what-is-child

        (animal ?name)

        (not (child-of ?name ?))

=>

        (printout t "What do you call the child of a " ?name "?")

        (assert (child-of ?name (read))))

Системата ще изиска въвеждане на име на животно за което няма достатъчно информация. След това ще се използват въведените данни за въвеждането на факт, който след това може да се използва от други правила.

(defrule are-lights-working

        (not (lights-working ?))

=>

        (printout t "Are the car's lights working (yes or no)?")

        (assert (lights-working (read))))

Конструкции

В CLIPS са дефинирани следните конструкции: defmodule, defrule, deffacts, deftemplate, defglobal, deffunction, defclass, definstances, defmessage-handler, defgeneric и defmethod. Всички конструкции се заграждат в кръгли скоби. Дефинирането на конструкция се различава от извикването на функция главно в ефекта. Функционалното извикване обикновено оставя средата на CLIPS непроменена (с някой изключения като например reset, clear или отварянето на файл). Дефинирането на конструкции цели да извърши някакви промени в средата чрез добавяне в базата от знания на CLIPS. За разлика от функциите, конструкциите никога не връщат стойност.

Конструкция defglobal
Чрез конструкцията defglobal могат да се дефинират глобални променливи, да им се задават стойности и да се достъпват в средата на CLIPS. Глобалните променливи могат да се достъпват като част от процеса на съпоставяне на образци. За задаване на стойност на променлива може да се използва функцията bind. При изпълнение на командата reset глобалните променливи възстановяват оригиналната си стойност. Същият резултат може да се постигне и чрез използването на bind без задаване на стойност. Това поведение може да се промени чрез използване на функцията set-reset-globals. Глобалните променливи могат да се премахнат чрез използване на командата clear или undefglobal. Ако глобалният елемент се наблюдава, при промяна на стойността на променливата ще се изведе съобщение.

Синтаксис:

(defglobal [<defmodule-name>] <global-assignment>*)

<global-assignment> ::= <global-variable> = <expression>

<global-variable> ::= ?*<symbol>*

Може да има множество конструкции defglobal и във всяка от тях могат да се дефинират произволен брой глобални променливи. Незадължителния <defmodule-name> показва модула в който се дефинират глобалните променливи. Ако не се зададе такъв, променливите ще се дефинират за текущия модул. Ако една глобална променлива вече е била дефинирана, новата дефиниция ще замести старата и променливата ще има стойност зададена чрез последната дефинираща конструкция. Ако при дефинирането на глобална променлива възникне грешка всички дефиниции на глобални променливи преди възникване на грешката ще продължат да съществуват.

Команди които работят с глобални променливи като ppglobals и undefglobal използват символно име на променливата без звездички (напр. трябва да се използва символа max когато е необходимо да се използва глобалната променлива ?*max*)

Глобалните променливи могат да се използват навсякъде където могат да се използват и локални променливи (с две изключения). Глобалните променливи не могат да се използват като променлива параметър за deffunction, defmethod или message-handler. Глобалните променливи не могат да се използват както локалните в лявата част на правилата за сжързване със стойност. Така например следното правило е грешно:

(defrule example
(fact ?*x*)

=>)

Следното правило обаче е правилно:

(defrule example

(fact ?y&:(> ?y ?*x*))

=>)

Не е необходимо това правило да се променя при промяна на стойността на ?*x* Например ако ?*x* е 4 и се добави факта fact3, то правилото не се удовлетворява. Ако стойността на ?*x* сега се промени на 2, правилото няма да е активно.

Пример:

(defglobal

?*x* = 3

?*y* = ?*x*

?*z* = (+ ?*x* ?*y*)

?*q* = (create$ a b c))

Printout
Функцията printout дава възможност за извеждане към устройство зададено, чрез логически име. Логическото име трябва да е зададено и устройството трябва да е подготвено за извеждане (напр. ако е файл, той трябва преди товада бъде отворен). За изпращане на изход към stdout за логическо име за задава t. Ако се използва логическо име nil, функцията printout не изпълнява никакви действия.

Синтаксис:

(printout <logical-name> <expression>*)

За извеждане в printout могат да се зададат произволен брой изрази. Всеки израз се изчислява и извежда (без интервали между всеки изведен израз). Символът crlf използван на мястото на <expression> предизвиква връщане на каретката и преминаване на нов ред и той може да се поставя на всяко място в списъка от изрази за извеждане. По подобен начин могатда се използват символите tab, vab и ff за извеждане съответно на табулация, вертикална табулация и form feed. Действието на тези специални символи може да се различава на различните операционни системи. Функцията printout пропуска кавичките около низове при извеждането им. Fact-address, instance-address и external-address могатда се изведат чрез printout. Функцията не връща стойност.

Пример:

CLIPS> (printout t "Hello there!" crlf)

Hello There!

CLIPS> (open "data.txt" mydata "w")

TRUE

CLIPS> (printout mydata "red green")

CLIPS> (close)

TRUE

CLIPS>

---------------------------------------------------------------------------

Задачи за самостоятелна работа:

1. (Както зад1 в ПРОЛОГ) Нека имаме определени следните три релации студент(фак. номер, име), предмет(номер на предмет, име), оценка(фак. номер, номер на предмет, оценка). Да се направи база, в която чрез факти да се опишат горните релации. Да се зададат следните вапроси:

  -Какво е името на студент, който има 6 по нещо. 

  -Какво е името на студент и името на предмет, по който има 6 

  -Има ли някой 6 по анализ и ТП. 

2. Напишете израза “10 + 4 * 19 – 35 / 12” в CLIPS и сравнете резултата с този на своите колеги.

